
A View on Views

This document assumes you have experience with Visual FoxPro tables, and creating SQL Select statements from within Visual FoxPro.

What’s a view?  It’s really nothing more than a query, created with an SQL Select statement, which is stored in a database container (DBC).  But the real advantage of a view is it can be updateable, meaning changes made to fields in a view can be written back to the original source table.

Beyond that basic definition, there are a few different kinds of views.  There are local views and remotes views.  There are parameterized views and non-parameterized views.  I’ll define each of these and show an example of the use of each.

Data Used

For all examples used in this document, I’ll be using six tables, Customers, OrdHead, OrdItems, Inventory, OrdServ and Services.  These files should be available with this document.  The use of the Customer table is pretty obvious.  OrdHead is a table of orders, with information that applies to the entire order.  OrdHead is related to Customers  by CustId, a field common to both tables.  OrdItems is a table of individual items on a order.  It relates to OrdHead by OrdHeadId.  Inventory is where we get the description and price for items in OrdItems.  OrdItems has an InvenId field to relate it to Inventory.  OrdServ is a list of services which apply to orders, and is related to OrdHead using OrdHeadId.  It is analogous to OrdItems.  Services is a lookup table for the types of services that may be added to orders.  OrdServ relates to this table on ServiceId.

The Plain Vanilla View

Let’s start with the simple first.  We’ll create a local, non-parameterized view.  This means the source table(s) for the view is a DBF, which we access natively, and there is either no filtering, or the conditions for the filter are static.  The DBF can be either a 2.x DBF or a VFP DBF.  Note also that even if the DBF files are located on a file server, you can still access them using a local view.  The difference between a local view and a remote view is how you access the data, not where that data is stored.  Because VFP can access DBF files natively, no matter where they are, a view using DBF files directly is a local view.

Let’s use the VFP view designer to build a view of  customers and a list of dates they’ve placed orders.  If you haven’t created a DBC yet, go ahead and do so.  I’ve called mine MyViews, but you can call yours whatever you want.  If you use the MyViews.DBC included with this document, be sure you don’t delete the view already there, because you’ll need it later.  If you have tables in another DBC that you would like to use, you need to open that DBC as well as the one with your views.  I prefer to keep my views in a separate DBC for reasons I’ll get into later.  Open the database designer.

As you would expect of VFP, there’s more than one way to create a view.  You can use the “New local view” button on the database designer toolbar, you can right-click on an empty portion of the DBC and select “New local view” from the pop-up menu, or you can select the “File” menu pad and then choose “New…”.  From there, you’ll get the [image: image1.png]Fie type

following dialog.

Dialog from the “File…New…” menu.
Select the “View” option and click on the “New file” button.

Define the Data Source

Any one of these techniques will open the view designer.  Since this is a new view, the view designer will ask you to define the source table(s) for the view.

Defining data source for a view
Note that you can use either tables (DBFs) or views as the source for a view.  For this document, we’ll just use tables, but there isn’t anything different you have to do to use a view, other than click on the “Views” option.  (If your tables aren’t part of a DBC, click on the “Other” button.)  I’ll select the Customers and Ordhead tables.

All open DBCs will be listed in the dropdown below “Database”.  If you need to use tables from another DBC, select that DBC from the dropdown, and the tables in that DBC will be listed under “Tables in database”.

Define the Join

Next, we need to define how these tables are related, and the type of join.  As you add any table after the first one, VFP will bring up a dialog asking for this information.

Defining a join
Define the Rest of the View

At this point, you are on the main View Designer screen.  It’s best if you define the rest of the parts of the view by selecting each tab of the View Designer’s pageframe in the order shown.

View Designer’s Main Screen
Select the fields from the data sources that you would like to be in your view.  You don’t have to restrict the fields you select to fields that the user can update in the view.  Some of them may be for display purposes only.  For this example, you should select the Customers.Custname, Ordhead.Ordno, Ordhead.Orddate, Ordhead.Ordheadid and Ordhead.Shipvia fields.

The next tab is the Join tab.  You only need to go to this tab if you want to change the join that you defined when you added data sources.

Next, define any filter conditions for the view.

Defining Filter Conditions
In the example above, I only want to view orders placed on or after January 1, 2001, so I’ve setup a filter condition that only includes order dates of January 1 or later.  You can additional filter conditions as well.

The next step is to define the order of the result.

Define the View’s Order
In the example above, I defined the order to be based on the customer name in ascending order.

Then, define any Group By clause for the view.  In this example, we won’t use a Group By.  

Next, we need to specify which fields in the view are updateable. 

Define Updateable Fields
It’s good practice to only have fields from a single table updateable.  (If you are defining a remote view that uses a client/server backend, chances are you aren’t allowed to have more than one table updated with a view.)

I strongly recommend you have a single field that uniquely identifies the record to be updated in the source table.  A local view will work if have multiple fields that uniquely identify each record, but it’s not a good idea.  Put a check mark in the column with the key icon to designate that key field or those fields.  Put a check mark in the column with the pencil icon for any fields that are updateable from the view.  Make sure you also check the box in the lower left corner that says “Send SQL updates”, or your changes will not be written back to the data source.  If the column or columns you have designated for your key identifies more than one record in the source, then when your changes are written back to the source, it will affect all those records.

As for the sections called “SQL Where clause includes” and “Update using”, I always leave those set to “Key and modified fields” and “SQL UPDATE”.  The on-line help has a good write-up of what these do.  Just look under “View Designer…Update Criteria Tab”.  I would like to add one caveat, however.  If you opt to update using “SQL Delete then Insert”, make sure that your primary key is incremented whenever you add or change a record in the view, and mark the primary key as updateable.  Otherwise, it deletes the record in the original table, but the TABLEUPDATE() will then fail because of a uniqueness violation.  Doing TABLEREVERT() will not bring back the deleted record in the source table.

Save this view, and call it CustOrds.  You can close the database designer window too.  Let’s open this view.  From the command window, type USE CustOrds.  This command will create a cursor with an alias of CustOrds, which you can use just as you would any cursor in Visual FoxPro.  Let’s BROWSE this view.

[image: image8.png]/ Microsoft Visual FoxPro

Ele Edt Yiew Jods

. 5|
custid

custname
custadr!
custaddi2
custey |

KIS |

Program Query Window _tielp

453 View Designer - View3

=]
ordheadd
custd
ordero

arddete
shipvia

~=lolx|
=l
]

it | don | Fitr | 8 | Group By UpdoteCieis | Miceloncus

Tape £.0 Fisname
A Taes Customerscusame
v Ocheadodhesdd
Bessikey v Ohesdardeno

Updste All

7 oS

v Oudheadorddate
Orchead shiavia

SOLWHERE clause ncludes
 Keyfelds only

€ Key and updatabe fields
 Key and modifed felds
€ Key andinestann.

Updateusing
" SQLDELETE then INSERT

 SQLUPDATE

(i

=lolx|

B
=]

afRel5

||
=% |8

=181]

View Desigrer

o |

EET




Browse of the new view
Notice that there’s only one record showing.  This is because only one order from our OrdHead table had an order date on or after January 1, 2001.  To confirm this, let’s browse the original OrdHead table.  First, you should know that opening a view, like running a regular SQL Select, will open all the source tables and leave them open.  So, all we have to do is SELECT the OrdHead table, which is already open.  Go ahead and browse this table.  Notice that all but one of the orders was placed in September 2000.

Changing Data in a View

You can use this view just as you would a table.  You can use fields from the view as ControlSources for objects on forms.  You can even use it for other queries.  One difference between tables and views, though, is that views must use optimistic buffering with either row or table buffering. When you make changes to fields in the view, those changes won’t appear in the source table(s) until you issue a TABLEUPDATE() command.  One thing you may notice is that fields in the view which are not marked as updateable may still be changed in the view.  These changes will not be written back to the source table.  If you use non-updateable fields as controlsources for things like textboxes, it’s important to make those objects disabled or read-only so the user doesn’t make changes and wonder why they’re not saved.  To test this, in the browse of the view above, change the date to 01/10/01 and the Shipvia field to UPS Red.  Close the browse and type TABLEUPDATE().  Select the Ordhead table an browse it.  Find order number 1010 and notice that the date is now 01/10/01, but the Shipvia is still FedEx.  This is because we designated the OrdDate field as updateable, but not the Shipvia field.

If you are familiar with the use of SQL Select commands, you may already know that when you run a SQL command, it will open any tables used by that command, and leave them open after it’s complete.  These tables are not buffered in any way.  The same is true for local views.  One thing you don’t want to do unless you have a very good reason to, is open the source tables ahead of time, and have them buffered.  If you do this, when you issue a TABLEUPDATE() command on the view, the changes from the view will be updated in the source tables, but you’ll need a second TABLEUPDATE() on the source table to actually save the changes written to that table.

Parameterized Views

Although it may be useful at some point to only see orders with date on or after January 1, 2001, hard-coding that into your view isn’t very useful.  It makes more sense to base this date on a variable.  A view that uses a variable in this way is called a “parameterized view”.  

Let’s turn the CustOrds view into a parameterized view.  Open the view designer and go to the Filter tab.  Change the contents of the “Example” column to say “?ldOrdDate” (without the quotes).  Make sure you move off of the Example column before you save your changes to the view.  Now go to the command window, and USE CustOrds.  You should have a dialog that looks like the one shown below.

Entering the value for a parameterized view
This dialog is asking for the value of the variable you defined in the filter.  If you had set this variable to a date before opening the view, you wouldn’t see this dialog.  You may also define the variable in the view designer without the question mark in front.  That question mark just tells VFP to show the above dialog if the variable hasn’t been defined.  If you do leave off the question mark, though, make sure you’ve defined the variable or you’ll get an error when you try run the query.

If necessary, you may open the view without running the query, and therefore, not need to define the variable until after you open the view.  To do this, use the NODATA clause when you issue the USE command.  You will then need to use the REQUERY() function to run the query on the selected view.

Types of Joins

Although I’ve made the assumption that you understand SQL Select statements, I think it’s import to have a quick review of the types of joins.  Up until now, we’ve only dealt with an inner join.  An inner join returns records where the field you are joining on has a match in each table.  So when we joined Customers with Ordhead on CustId, that returned only customers who had orders.  See the diagram below for a visual example of this.

Diagram of what an inner join looks like
What if we want to see all customers, regardless of whether or not they’ve placed an order, plus any order information if they have placed an order.  We need what’s called an “outer join”.  There are Left Outer Joins, Right Outer Joins, and Full Outer Joins.  The difference between a left outer join vs. a right outer join is which table is on the right of the join equation vs. which is on the right.  To perform a left outer join from Customers into OrdHead, we’d use a SQL select like:

Select Customers.CustName, OrdHead.OrdNo ;

   From Customers ;

      Left outer join OrdHead ;

         On Customers.CustId = OrdHead.CustId

What that would look like in a diagram is this:

Diagram of a left outer join
As you would expect, in a right outer join, the entire right circle would be filled in, but for the left circle, only the part that intersects with the right circle would be filled in.  For a full outer join, we would get all records from both tables whether there were any matches or not.  The diagram would have both circles completely filled in.

One thing to be aware of with outer joins is the possibility of null values in your result.  If there is no match in the related table, you will get a null in the field.  Handling nulls can be tricky, so I usually use the NVL() function to define what value I’d like to get instead of a null when there’s no match in the related table.

Defining Views In Code 

Now that I’ve shown you how to use the View Designer, I’ll let you know that I never use it.  That’s right, I never use the View Designer.  The reason for this is that the View Designer is fairly limited, and has some other problems.  To demonstrate one of these problems let’s modify the CustOrds view again. Open the View Designer and right click on the top half of the View Designer.  (The half with the tables, but don’t right click on one of the tables.)  Select “Add Table”, and add the OrdItems, Inventory, OrdServ and Services tables.  Notice that the relationships shown in the top portion of the View Designer look fine.  Now save this view as CustOrds2.  Try to modify it again, and you’ll find that VFP blows up with the message “SQL: Column ‘ORDHEADID’ is not found”.  This is what is called two unrelated siblings, and the View Designer can’t handle this.  Furthermore, you can’t even get the view open to fix it, and must delete this view and start all over.

Another problem is that when your view contains data from more than one table, the View Designer may specify more than one table as updateable, even though you’ve only defined fields from one of the tables as updateable.

The third problem is the syntax the View Designer uses for outer joins.  I like to put the “ON” condition right after specifying the table I’m joining.  For example:

Select Customer.CustName, ;

       nvl(OrdHead.OrdNo,0000000000) as OrdNo, ;

       nvl(OrdItems.QtyOrd) as QtyOrd ;

   From Customer ;

      Left outer join OrdHead ;

         On Customer.CustId = OrdHead.CustId ;

      Left outer join OrdItems ;

         On OrdHead.OrdHeadId = OrdItems.OrdHeadId

I think this is a much clearer syntax, because you see the join condition grouped with the file you’re joining.

Another valid syntax, and the one which the view designer uses, is to specify all the join conditions after you specify the files in the join.  You specify the join conditions in reverse order from the way you specified the files.  So, using this syntax, the above example would look like this:

Select Customer.CustName, ;

       nvl(OrdHead.OrdNo,0000000000) as OrdNo, ;

       nvl(OrdItems.QtyOrd) as QtyOrd ;

   From Customer ;

      Left outer join OrdHead ;

      Left outer join OrdItems ;

         On OrdHead.OrdHeadId = OrdItems.OrdHeadId ;

         On Customer.CustId = OrdHead.CustId 

You can verify this with the View Designer by clicking on the “SQL” button of the toolbar.  This shows you the SQL code generated by the View Designer.

GenDBC to the Rescue

The best way to see generate an example of what it looks like to generate your views in code is to use GenDBC, a tool that ships with VFP.  You can find GenDBC in the Tools\GenDBC directory under the directory where you have VFP installed.  GenDBC will use the current DBC if there is one.  If you don’t have a DBC open, it will ask for the DBC.  GenDBC also has an optional parameter to tell it what file to send the output code to.

Assuming you still have the DBC open, go ahead and run GenDBC with a parameter of “TEST”.  Modify the Test.prg you just created.  You’ll notice that it has code to create the DBC as well as the code to create the view.  Creating the view consists of a SQL Select command, plus code to set properties for the view, and code to set properties for each of the fields in the view.  At this point, you probably realize it’s not very difficult to create any view you want in code.  In fact, you can create views that aren’t even possible with the view designer, like one that has two or more unrelated siblings.

Another Alternative

I consider repeating the code to set the view properties for each view and each field to be a bad programming example.  Repeated code is something to be avoided.  I’ve created a SetViewProperties procedure that you can call to set these properties.  This procedure is in the MakeView.prg that should be available with this document, and from my website at www.peisch.com.  If you examine this procedure, you’ll notice that it also sets all the field properties in a loop instead of repeating that code for each field.  There are some requirements to use this function, and those are documented at the beginning of the program.  I also have a utility option from within my programs that will delete the DBC and call the code that regenerates it, so that users can easily do this when necessary when I suspect a problem.

You may recall that near the beginning of this document, I mentioned that I prefer to keep views in a separate DBC.  To demonstrate why, create a new view that just pulls all fields from the customers table.  Save this view as CustView.  Now modify the structure of the customers table.  Add a new field called CustZip4, a 4 character field.  Save your changes to the data structure.  Now try to USE the view.  You’ll get an error message “Base table fields have been changed and no longer match the view fields. View field properties cannot be set.”  Although you can modify the view, and make the necessary changes to fix this problem, as you can imagine, it could be a challenge to find all the views that use this table in a system of any size.

If you’ve written code to create your views, searching for this table in that code is easy.  If you’ve used select * from the table, you don’t even need to change the code.  You just need to run the code to regenerate the DBC.  I prefer to have the code that creates the DBC and all the views only run when the DBC is missing, and this check is run every time I start my program.  So, all I have to do is erase the DBC and restart the program, and all my views are fine.

If you do create your views in code, and you are using tables from a DBC other than the one for your views, make sure you reference that DBC name in your select statement.  For example, assuming that all my tables are in a DBC called OrderSystem, my code might look like this:

Create SQL view “CustOrds” as ;

  Select Customer.CustName, ;

       nvl(OrdHead.OrdNo,0000000000) as OrdNo, ;

       nvl(OrdItems.QtyOrd) as QtyOrd ;

   From OrderSystem!Customer ;

      Left outer join OrderSystem!OrdHead ;

         On Customer.CustId = OrdHead.CustId ;

      Left outer join OrderSystem!OrdItems ;

         On OrdHead.OrdHeadId = OrdItems.OrdHeadId

Assuming you are familiar with SQL statements, you’ll notice that this is exactly the same syntax you’d use to specify a DBC name in a SQL statement.

Index Tags on a View

Unlike a table, views cannot have permanent tags that you define in one place.  Each time you open a view, you must re-create any index tags needed.  This changes in VFP7 with the addition of DBC events.  There, you can put any index tags for a view in that view’s AfterOpenTables method.  

One thing that’s important to realize is that you can only create index tags on a view if it is in row buffering mode.  Row buffering is the default buffering mode used when you open a view, so if you want tags and table buffering, just create your index tags first, and then change the view’s buffer mode to table buffering.  If you need to add a tag to a view that’s in table buffering mode, change it to row buffering, build your tag, then change the buffer mode back to table buffering.

Multi-user Considerations

When a program opens a view, there is a brief period while the view definition is being retrieved that the DBC is locked.  This means that if another user tries to open a view at that time, they’ll get an error.  The easiest way to avoid this problem is to keep the DBC on each user’s local hard drive.  Of course, you can only do this if your views are in a separate DBC, otherwise, tables in the DBC couldn’t be shared.

There are two ways to accomplish this goal.  One is to have the code that creates the DBC and views create them directly on the local hard disk.  If you choose to use this technique, I recommend that you always recreate the DBC when the program starts, in case there have been changes in any of the views.  That way you won’t have to remember to erase the DBC on each user’s system when you make changes, or have to remember to ask each user to run your utility to recreate the views.

The second technique has the code that creates the DBC and views create them on the server, but when a user starts the system, a new copy of the DBC, DCT and DCX files are always copied to the local hard drive.  With this technique, when you make any view changes, you must still remember to erase the DBC on the server or run the utility to recreate the views, but you only have to do that once.

Regardless of which technique you use to put the database for the views on the local hard drive, you should add that location to the VFP path, so that your program can find those views without having to specify a full path.

Putting Views to Use

So, let’s look at a practical example for using a parameterized view.  I’m going to create a form that shows order header information, plus a grid with items for the current order.

For this section we’ll use a view that I’ve already defined, called OrderItemsView in the DBC MyView that’s included with this document.  In case it’s missing, you can create it as follows:  Create a new local view.  Include all the fields from OrdItems, plus the InvenNum and Descrip fields from Inventory.  For a filter expression, use OrdItems.OrdHeadId = ?lCurOrdHeadId.  Use InvenNum as the field to order by.  Mark the OrdItemId field as the key field and make all the fields from OrdItems updateable, including the OrdItemId field.  Don’t forget to check the “Send SQL updates” box.  Save the view.

Setting up the Data Environment

Create a new form called Orders, or use the one that comes with this document.  If you’re creating your own, right-click on the form and select “Data environment”.  Add the OrdHead, Customers and Inventory tables, then the OrderItemsView.  Create a relationship between OrdHead and Customers by clicking on CustId in OrdHead and dragging over to the CustId field or index in the Customers file.  Right-click on the view and select “Properties”.  Change its NoDataOnLoad property to .T.  This setting will keep the view from expecting lCurOrdHeadId to have a value before we have a chance to set it.  Create a relationship between the view and the Inventory table by clicking on the InvenId field in the view and dragging it to the InvenId field or index of the Inventory table.  Set the order of the OrdHead table to OrderNo by right-clicking on it, selecting “Properties”, and selecting OrderNo for the Order property.

As an alternative, you can setup the data environment in code.  If you would like to retain the advantages of having tables in the visual designer, but have the added control of opening the tables in code, add the tables and the view to the data environment as I’ve described above, but then right-click on the data environment in a section without any tables.  Select “Properties” and change the AutoOpenTables property to .F.  If you also would like to close the tables manually, set the AutoCloseTables properties to .F. too.  Then, put the following code in the form’s Load method:

Use Customers in 0 order CustId

Select 0

Use OrdHead order OrderNo

Set relation to CustId into Customers

Use Inventory in 0 order InvenId

Open database MyViews

Select 0

Use OrderItemsView Nodata

If you want to close the tables in code, don’t forget to add a CLOSE ALL to the form’s Destroy or Unload method.

Setting up the Form

Put the Customers.CustName field on the form, along with all the fields from OrdHead except OrdHeadId and CustId.  Then add a grid with four columns, using OrderItemsView as its RecordSource.  Use the Inventory.InvenNum for the first column, the Inventory.Descrip for the second column, and each of the quantities in OrderItemsView for the remaining three columns.  I’ve also added an exit button, and forward and back buttons.  I use a special method, called from the form’s Refresh method to change the variable lCurOrdHeadId as we move the pointer in the OrdHead table, and requery the view.  Look at the form Orders.scx included with this document, particularly in the Click method for the forward and back buttons for details.

As you move from record to record in the form, notice how the items shown are only the items for the current order. This is all handled by the parameterized view.  We often can accomplish the same thing by setting the LinkMaster and RelationalExpr properties of the grid, but I’ve had mixed results with that technique.  Also, there are situations where using a table directly as the recordsource for a grid causes a “Record is in use by another” error.  Using a parameterized view is definitely the preferred method.  Also, we’ve combined data from two tables into this single view, so that we can easily display normalized data.

Accessing Remote Data

You can use what’s called a remote view to access data through an ODBC connection (Open DataBase Connectivity).  

Setting Up an ODBC Connection

Setting up an ODBC connection can get pretty involved, but for this discussion, we’ll keep things simple.  We’re just going to define a connection to the same VFP tables we’ve been using.

If you’re running Windows 9x or NT 4, you’ll find an icon for ODBC Data Source on the Control Panel.  In Windows 2000, you must select the “Administrative Tools” option from the Control Panel, then select “Data Sources (ODBC)”.  (There’s an alternative way to define an ODBC Data Source that you may find more convenient.  I’ll get to that shortly.)

Chances are, you’ll want a System DSN, so that multiple users can share the same connection, so let’s define our new connection as a System DSN.  Click on the “System DSN” tab.

[image: image12.png]User DN SysemDSH | i 05N | e | Tracing | Cornestion P | About

System Data Sources:

Name

Diiver

"WEEN ODEC Diver

An DDBC System data source stores information about how to conmect to
the indicated deta provider. A System data source is visble o al users
on this machine, including NT services.

oK

Help




Defining a new System DSN
Click on the Add button.  If you installed the ODBC drivers when you installed VFP, you should have that driver listed, as shown below.

Selecting the VFP driver for ODBC
Click Finish.  You will then get a screen like this one:

Defining the ODBC Connection
You can enter what you want for the Data Source Name and Description.  Select either Visual FoxPro database (DBC) or Free Table directory, depending on whether or not the tables you want to use are part of a DBC.  Finally, enter the directory for the source data, and click OK.

Once you have an ODBC connection defined, you may create the connection information in the DBC you’ll be using for your remote view.  This is optional, but creating a connection in the DBC allows you to specify some options you wouldn’t have otherwise.  Using the same DBC for your local and remote views is fine.  Define a connection within your DBC by right-clicking on a blank area of the DBC in the database designer, or by selecting  the File…New...Connection option.  You’ll see a dialog something like this:

Defining connections for remote views
Select “New” if you need to add a new connection.  You then get a screen like this:

Defining the details for a connection
Select the ODBC connection you defined for the “Data source”.  That’s all you’ll need for this discussion.  Press OK and you’re ready to create a remote view.

You may have noticed the “New Data Source” button on this screen.  That is an alternative way of defining the ODBC Data Source.  So, you can do it all from this one screen, which is a lot easier than tracking down where your operating system keeps the ODBC Data Source definitions item.

It’s also possible to define your ODBC connection in code by using an API call, but that’s more than I want to get into here.  The advantage is you can define the connection and remove it in code, without having to rely on someone to set this up ahead of time.

Creating a Remote View

You create a remote view in much the same way you would a local view.  Just choose “Remote view” instead.  If you defined a connection in the DBC, you’ll see that connection listed.

Selecting data source or connection for a local view
If you didn’t define a connection in the DBC, choose “Available data source”, and you’ll see the ODBC data sources available.

From this point on, defining the remote view is no different from defining a local view, except that your choice of source tables are only those defined by your connection or ODBC data source.  Go ahead a create a remote view of all customers.  Then, use GenDBC to generate code to create this DBC and view.  If you defined a connection, you’ll  code like this near the top:

CREATE CONNECTION VFPCONNECTION ; 

   DATASOURCE "RemoteViews" ; 

   USERID "" ; 

   PASSWORD ""

****

DBSetProp('VFPCONNECTION', 'Connection', 'Asynchronous', .F.)

DBSetProp('VFPCONNECTION', 'Connection', 'BatchMode', .T.)

DBSetProp('VFPCONNECTION', 'Connection', 'Comment', '')

DBSetProp('VFPCONNECTION', 'Connection', 'DispLogin', 1)

DBSetProp('VFPCONNECTION', 'Connection', 'ConnectTimeOut', 15)

DBSetProp('VFPCONNECTION', 'Connection', 'DispWarnings', .F.)

DBSetProp('VFPCONNECTION', 'Connection', 'IdleTimeOut', 0)

DBSetProp('VFPCONNECTION', 'Connection', 'QueryTimeOut', 0)

DBSetProp('VFPCONNECTION', 'Connection', 'Transactions', 1)

DBSetProp('VFPCONNECTION', 'Connection', 'Database', '')

As you can see, being able to define your connection in code offers great flexibility.  As for the definition of the view itself, you’ll notice the similarity to the code to create a local view. 

CREATE SQL VIEW "CUSTOMERSVIEW" ; 

   REMOTE CONNECT "VFPConnection" ; 

   AS SELECT * FROM customers Customers

The difference is it has the addition of a “Remote connect…” clause.  The name specified in this clause is either the ODBC data connection, or the connection defined in your DBC.  The syntax is exactly the same in either case.

With remote views, you are no longer limited to querying FoxPro data, you can query data from any source that has an ODBC driver available.  This includes client/server backends like SQL Server and Oracle, as well as other tools like Access and Delphi.  The beauty of this is, once you have a remote view to any data source, you can use that data in VFP just as if it was native data, and you can even write changes back to the original source!

The FoxPro Wiki

If you haven’t discovered Wikis yet, they are a wonderful repository of knowledge.  Think of a wiki as an interactive knowledge base.  The FoxPro wiki has discussions on the use of remote views, looking at the pros and cons, and comparing them to SQL pass-through (SPT), which is another alternative for accessing remote data in VFP.  Visit links at http://fox.wikis.com/wc.dll?Wiki~RemoteViews and http://fox.wikis.com/wc.dll?Wiki~MoreOnRemoteViews  

There’s also a page on tips for local views at http://fox.wikis.com/wc.dll?Wiki~LocalView_TipsTricks&Gotchas
� Thanks go to Twila Miller for teaching me about this trick! You can reach Twila at tamiller@tmdesignware.com

